HEWLETT hp PACKARD

1000 N.E. Circle Blvd., Corvallis, OR 97330

For additional sales and service information contact your local Hewlett-Packard Sales Office or Call 800/648-4711. Excludes Alaska and Hawaii. (In Nevada call 800/992-5710.)

00034-90022

Printed in U.S.A.

HEWLETT-PACKARD

HP-34C APPLICATIONS

NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

HEWLETT hp PACKARD

HP-34C

Applications

May 1979

00034-90022

Printed in U.S.A.

Introduction

Congratulations on owning an HP-34C. We know you will be pleased with its quality, versatility and ease of use. Its programmable capability and powerful built-in functions combined with Continuous Memory make it a uniquely useful calculator.

This applications book is designed to help you get the best from your calculator. It provides programs to give you answers to "real world" problems, as well as games and other programs of general interest.

The programs include interesting techniques which you may find useful in writing your own software. We are confident you will find this book useful and we welcome your comments and suggestions.

Contents

A Word About Program Usage

Each program is accompanied by a brief description of the problem, the applicable equations, a listing of program keystrokes, a set of instructions for using the program and one or more example problems, showing the actual keystrokes required for the solution.

Program listings are provided in the following format: (This example is from *Curve Fitting,* the first program in this book.)

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000-	[GTO] 6	026- 22 6
[h] [LBL] [A]	001-25, 13, 11	[f] [Σ+]	027- 14 74
[f] CLEAR [REG]	002- 14 33	[GTO] 9	028- 22 9
[f] [FIX] 2	003-14, 11, 2	[h] [LBL] 6	029-25, 13, 6
[ENTER↑]	004- 31	[g] [Σ-]	030- 15 74
[ENTER↑]	005- 31	[h] [CF] 2	031-25, 61, 2
6	006- 6	[GTO] 9	032- 22 9
[+]	007- 51	[h] [LBL] [B]	033-25, 13, 12

The leftmost column, headed KEY ENTRY, shows the keys which must be pressed to enter the program into program memory. All the key designations are identical with the way they appear on your keyboard. The second column, headed DISPLAY, shows the appearance of the display on the calculator as you key in the program. The first three numerals on the left are the line number, followed by a dash, then the numeric keycode corresponding to the keystrokes in the KEY ENTRY column. Storage register contents are shown at the end of the program listing.

The USER INSTRUCTIONS form is your guide to using the program to solve your own problem. The first column, labeled STEP, gives the instruction step number. Steps are executed in sequential order except where otherwise noted. the INSTRUCTIONS column gives instructions and comments concerning the operations to be performed. The INPUT DATA/UNITS column specifies the input data to be supplied, and, if applicable, the units of the data. Data input keys consist of 0 thru 9 and decimal point (the numeric keys), [EEX] (enter exponent) and [CHS] (change sign). The KEYS column specifies the keys to be pressed after keying in the corresponding input data. The OUTPUT DATA/UNITS column specifies intermediate and final outputs and, where applicable, their units.

The form is illustrated below for the same program, *Curve Fitting.*

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Select the type of curve fit:			
	Exponential	1	[A]	1.00
	Logarithmic	2	[A]	2.00
	Power	3	[A]	3.00
3	Input x_i value	x_i	[ENTER↑]	
	and y_i value.	y_i	[R/S]	i
	(Repeat step 3 for all			
	data points.)			
4	Calculate regression			
	coefficients.		[B]	a
			[R/S]	b
	and coefficient of determin-			
	ation.		[R/S]	r^2

Step 1 requires you to key in the program. Switch the HP-34C to PRGM mode, press the keys [f] CLEAR [PRGM] and key in the program steps as shown on the complete listing. Then switch the calculator to RUN mode and proceed with the USER INSTRUCTIONS.

Step 2 asks you to select the type of curve fit desired and input a corresponding code number. Key in the desired code number and press [A] Step 3 requests input of pairs of x- and y-values. Each x-value is keyed in and [ENTER↑] is pressed. Then the y-value is keyed in and [R/S] is pressed. This procedure is repeated until all pairs of values have been input.

Step 4 calculates the regression coefficients. The user presses [B] and sees the regression coefficient, a, displayed. Pressing [R/S] displays coefficient b. Press [R/S] to display r^2, the coefficient of determination.

Mathematics

Curve Fitting

Your HP-34C calculator is equipped with a powerful built-in function, linear regression, [L.R.], which quickly and conveniently fits data to a straight line. (Refer to your HP-34C Owner's Handbook and Programming Guide for full details.)

This capability is used here in a program to fit data to other types of curves:

1. Exponential curves; $y = ae^{bx}$ $(a > 0)$
2. Logarithmic curves; $y = a + b \ln x$
3. Power curves; $y = ax^b$ $(a > 0)$

which may be transformed to the general linear form $Y = A + bX$.

The regression coefficients a and b are found by solving the following system of linear equations.

$$\begin{bmatrix} n & \Sigma X_i \\ \Sigma X_i & \Sigma X_i^2 \end{bmatrix} \begin{bmatrix} A \\ b \end{bmatrix} = \begin{bmatrix} \Sigma Y_i \\ \Sigma (Y_i X_i) \end{bmatrix}$$

The relations of the variables are defined as the following:

Regression	A	X_i	Y_i	Code
Exponential	ln *a*	x_i	$\ln y_i$	1
Logarithmic	*a*	$\ln x_i$	y_i	2
Power	ln *a*	$\ln x_i$	$\ln y_i$	3

The coefficient of determination is:

$$r^2 = \frac{A\Sigma Y_i + b\Sigma X_i Y_i - \frac{1}{n}(\Sigma Y_i)^2}{\Sigma(Y_i^2) - \frac{1}{n}(\Sigma Y_i)^2}$$

The type of curve fit to be run is determined before data input begins by inputting the code number.

The coefficient of determination indicates the quality of fit achieved by the regression. Values of r^2 close to 1.00 indicate a better fit than values close to zero. The regression coefficients a and b define the curve generated, according to the equations at the beginning of this discussion.

Exponential Curve Fit
Code = 1



Logarithmic Curve Fit
Code = 2



Power Curve Fit
Code = 3



Remarks:

- The program applies the least squares method, either to the original equations (logarithmic curve) or to the transformed equations (exponential curve and power curve).
- Negative and zero values of x_i will cause a machine error for logarithmic curve fits. Negative and zero values of y_i will cause a machine error for exponential curve fits. For power curve fits both x_1 and y_1 must be positive, non-zero values.
- As the differences between x and/or y values become small, the accuracy of the regression coefficients will decrease.
- During operation of the program all storage registers are cleared. Any data stored in extra registers will therefore be destroyed.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL A	001-25, 13, 11
f CLEAR REG	002- 14 33
f FIX 2	003-14, 11, 2
ENTER↑	004- 31
ENTER↑	005- 31
6	006- 6
+	007- 51
STO f I	008-23, 14, 23
g R↓	009- 15 22
h SF 0	010-25, 51, 0
h SF 1	011-25, 51, 1
GTO f I	012-22, 14, 23
h LBL 7	013-25, 13, 7
h CF 1	014-25, 61, 1
GTO 9	015- 22 9
h LBL 8	016-25, 13, 8
h CF 0	017-25, 61, 0
h LBL 9	018-25, 13, 9
R/S	019- 74
h F? 0	020-25, 71, 0
f LN	021- 14 1
x⇄y	022- 21
h F? 1	023-25, 71, 1
f LN	024- 14 1
h F? 2	025-25, 71, 2

KEY ENTRY	DISPLAY
GTO 6	026- 22 6
f Σ+	027- 14 74
GTO 9	028- 22 9
h LBL 6	029-25, 13, 6
g Σ-	030- 15 74
h CF 2	031-25, 61, 2
GTO 9	032- 22 9
h LBL B	033-25, 13, 12
h L.R.	034- 25 6
h F? 0	035-25, 71, 0
g e^x	036- 15 1
R/S	037- 74
x⇄y	038- 21
R/S	039- 74
h r	040- 25 5
g x^2	041- 15 3
h RTN	042- 25 12
h LBL 0	043-25, 13, 0
h SF 2	044-25, 51, 2
GTO 9	045- 22 9
h LBL 1	046-25, 13, 1
h F? 1	047-25, 71, 1
f LN	048- 14 1
h $\hat{y}$	049- 25 4
h F? 0	050-25, 71, 0
g e^x	051- 15 1

REGISTERS			I Code + 6
R_0 n	R_1 ΣX_i	R_2 ΣX_i^2	R_3 ΣY_i
R_4 ΣY_i^2	R_5 $\Sigma X_i Y_i$	R_6—R_9 Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	Select the type of curve fit:			
	Exponential	1	A	1.00
	Logarithmic	2	A	2.00
	Power	3	A	3.00
3	Input x_i value and y_i value.	x_i	ENTER↑	
		y_i	R/S	i
	(Repeat step 3 for all data			
	points)			
4	Calculate regression coef-			
	ficients		B	a
			R/S	b
	and coefficient of determin-			
	ation.		R/S	r^2
5	Make projection of new $\hat{y}$ for			
	a known x value.	x	GSB 1	$\hat{y}$
	(Repeat step 5 for all x			
	values of interest.)			
6	Error Deletion:			
	Erroneous inputs at step 3 may		GSB 0	
	be corrected by pressing	x_i err	ENTER↑	
	GSB 0 and reinputting the	y_i err	R/S	$i-1$
	erroneous data. Then return to			
	step 3 and enter the correct			
	data.			

Example 1:

(Exponential, Code = 1)

x_i	.72	1.31	1.95	2.58	3.14
y_i	2.16	1.61	1.16	.85	0.5

Solution:

$a = 3.45$, $b = -0.58$

$y = 3.45\, e^{-0.58x}$

$r^2 = 0.98$

Keystrokes:	**Display:**	
1 [A]	*1.00*	
.72 [ENTER↑] 2.16 [R/S]		
1.31 [ENTER↑] 1.61 [R/S]		
1.95 [ENTER↑] 1.16 [R/S]		
2.58 [ENTER↑] .85 [R/S]		
3.14 [ENTER↑] .5 [R/S]		
[B]	*3.45*	a
[R/S]	*-0.58*	b
[R/S]	*0.98*	r^2
1.5 [GSB] 1	*1.44*	$\hat{y}$

Example 2:

(Logarithmic, Code = 2)

x_i	3	4	6	10	12
y_i	1.5	9.3	23.4	45.8	60.1

Solution:

$a = -47.02$, $b = 41.39$

$y = -47.02 + 41.39 \ln x$

$r^2 = 0.98$

For $x = 8$, $\hat{y} = 39.06$

For $x = 14.5$, $\hat{y} = 63.67$

Example 3:

(Power, Code = 3)

x_i	10	12	15	17	20	22	25	27	30	32	35
y_i	0.95	1.05	1.25	1.41	1.73	2.00	2.53	2.98	3.85	4.59	6.02

Solution:

$a = .03$, $b = 1.46$

$y = .03x^{1.46}$

$r^2 = 0.94$

For $x = 18$, $\hat{y} = 1.76$

For $x = 23$, $\hat{y} = 2.52$

Hyperbolic Functions

This program calculates the hyperbolic functions and their inverses with special algorithms that provide excellent accuracy. Calculation of the hyperbolic functions, as defined below, will yield errors in many instances for small arguments (i.e., $x < 0.5$) due to roundoff errors. However, this program has been specially written to avoid these errors and should provide accurate answers to better than eight significant figures.

Equations:

Hyperbolic Functions

$$\sinh x = \frac{e^x - e^{-x}}{2} \qquad \operatorname{csch} x = \frac{1}{\sinh x} \quad (x \neq 0)$$

$$\cosh x = \frac{e^x + e^{-x}}{2} \qquad \operatorname{sech} x = \frac{1}{\cosh x}$$

$$\tanh x = \frac{e^x - e^{-x}}{e^x + e^{-x}} \qquad \coth x = \frac{1}{\tanh x} \quad (x \neq 0)$$

Inverse Hyperbolic Functions

$$\sinh^{-1}x = \ln\left[x + (x^2+1)^{1/2}\right]$$

$$\cosh^{-1}x = \ln\left[x + (x^2-1)^{1/2}\right] \quad x \geq 1$$

$$\tanh^{-1}x = \tfrac{1}{2}\ln\left[\frac{1+x}{1-x}\right] \quad x^2 < 1$$

$$\operatorname{csch}^{-1}x = \sinh^{-1}\left[\frac{1}{x}\right] \quad x \neq 0$$

$$\operatorname{sech}^{-1}x = \cosh^{-1}\left[\frac{1}{x}\right] \quad 0 < x \leq 1$$

$$\coth^{-1}x = \tanh^{-1}\left[\frac{1}{x}\right] \quad x^2 > 1$$

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	000–	GSB 2	029– 13 2
h LBL A	001–25, 13, 11	+	030– 71
h CF 1	002 25, 61, 1	h LBL 7	031–25, 13, 7
h CF 0	003–25, 61, 0	h F? 1	032–25, 71, 1
STO f I	004–23, 14, 23	h 1/x	033– 25 2
g R↓	005– 15 22	R/S	034– 74
GTO f I	006–22, 14, 23	h LBL 0	035–25, 13, 0
h LBL 4	007–25, 13, 4	ENTER↑	036– 31
h LBL 5	008–25, 13, 5	h ABS	037– 25 34
h LBL 6	009–25, 13, 6	·	038– 73
h SF 1	010–25, 51, 1	5	039– 5
g DSE	011– 15 23	f x≷y	040– 14 41
g DSE	012– 15 23	GTO 8	041– 22 8
g DSE	013– 15 23	g R↓	042– 15 22
GTO f I	014–22, 14, 23	g e^x	043– 15 1
h LBL 1	015–25, 13, 1	f LN	044– 14 1
h SF 0	016–25, 51, 0	g x=0	045– 15 71
GSB 0	017– 13 0	GTO 9	046– 22 9
h CF 0	018–25, 61, 0	1	047– 1
GTO 7	019– 22 7	h LST x	048– 25 0
h LBL 2	020–25, 13, 2	−	049– 41
GSB 2	021– 13 2	h LST x	050– 25 0
GTO 7	022– 22 7	g R↓	051– 15 22
h LBL 3	023–25, 13, 3	+	052– 71
STO 0	024– 23 0	+	053– 71
h SF 0	025–25, 51, 0	CHS	054– 32
GSB 0	026– 13 0	·	055– 73
h CF 0	027–25, 61, 0	5	056– 5
RCL 0	028– 24 0	ENTER↑	057– 31

KEY ENTRY	DISPLAY
f R↑	058- 14 22
+	059- 71
+	060- 51
×	061- 61
h RTN	062- 25 12
h LBL 9	063-25, 13, 9
x⇄y	064- 21
h RTN	065- 25 12
h LBL 8	066-25, 13, 8
g R↓	067- 15 22
÷	068- 71
h LST X	069- 25 0
h LBL 2	070-25, 13, 2
g eˣ	071- 15 1
h LST X	072- 25 0
CHS	073- 32
g eˣ	074- 15 1
h F? 0	075-25, 71, 0
CHS	076- 32
+	077- 51
2	078- 2
÷	079- 71
h F? 0	080-25, 71, 0
×	081- 61
h RTN	082- 25 12
h LBL B	083-25, 13, 12
h CF 0	084-25, 61, 0
STO f I	085-23, 14, 23
g R↓	086- 15 22

KEY ENTRY	DISPLAY
GTO f I	087-22, 14, 23
h LBL 4	088-25, 13, 4
h LBL 5	089-25, 13, 5
h LBL 6	090-25, 13, 6
h 1/x	091- 25 2
g DSE	092- 15 23
g DSE	093- 15 23
g DSE	094- 15 23
GTO f I	095-22, 14, 23
h LBL 1	096-25, 13, 1
GSB 4	097- 13 4
g x²	098- 15 3
ENTER↑	099- 31
ENTER↑	100- 31
1	101- 1
+	102- 51
f √x	103- 14 3
1	104- 1
+	105- 51
÷	106- 71
+	107- 51
GSB 0	108- 13 0
h F? 1	109-25, 71, 1
CHS	110- 32
h RTN	111- 25 12
h LBL 3	112-25, 13, 3
GSB 4	113- 13 4
ENTER↑	114- 31
1	115- 1

KEY ENTRY	DISPLAY
x⇄y	116- 21
−	117- 41
÷	118- 71
ENTER↑	119- 31
+	120- 51
GSB 0	121- 13 0
h F? 1	122-25, 71, 1
CHS	123- 32
2	124- 2
÷	125- 71
h RTN	126- 25 12
h LBL 2	127-25, 13, 2
1	128- 1
−	129- 41
ENTER↑	130- 31
ENTER↑	131- 31
ENTER↑	132- 31
2	133- 2
+	134- 51
×	135- 61
f √x	136- 14 3
+	137- 51
h LBL 0	138-25, 13, 0
ENTER↑	139- 31
ENTER↑	140- 31

KEY ENTRY	DISPLAY
1	141- 1
f x>y	142- 14 51
GTO 0	143- 22 0
+	144- 51
f LN	145- 14 1
h RTN	146- 25 12
h LBL 0	147-25, 13, 0
+	148- 51
f LN	149- 14 1
h LST X	150- 25 0
1	151- 1
−	152- 41
g x≠0	153- 15 61
÷	154- 71
×	155- 61
g x=0	156- 15 71
x⇄y	157- 21
h RTN	158- 25 12
h LBL 4	159-25, 13, 4
h CF 1	160-25, 61, 1
g x<0	161- 15 41
h SF 1	162-25, 51, 1
h ABS	163- 25 34
ENTER↑	164- 31

REGISTERS		I Control
R_0 x	R_1—R_5 Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	For hyperbolics go to step 3,			
	for inverse hyperbolics go to			
	step 4.			
	HYPERBOLIC FUNCTIONS:			
3	Key in the argument and the			
	code:	x	[ENTER↑]	
	• hyperbolic sine (code= 1)	1	[A]	sinh x
	• hyperbolic cosine (code= 2)	2	[A]	cosh x
	• hyperbolic tangent (code= 3)	3	[A]	tanh x
	• hyperbolic cosecant			
	(code= 4)	4	[A]	csch x
	• hyperbolic secant (code= 5)	5	[A]	sech x
	• hyperbolic cotangent			
	(code= 6)	6	[A]	coth x
	INVERSE HYPERBOLIC			
	FUNCTIONS:			
4	Key in the argument and			
	the code:	x	[ENTER↑]	
	• inverse hyperbolic sine			
	(code= 1)	1	[B]	$\sinh^{-1}x$
	• inverse hyperbolic cosine			
	(code= 2)	2	[B]	$\cosh^{-1}x$
	• inverse hyperbolic tangent			
	(code= 3)	3	[B]	$\tanh^{-1}x$
	• inverse hyperbolic cosecant			
	(code= 4)	4	[B]	$\text{csch}^{-1}x$
	• inverse hyperbolic secant			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
	(code= 5)	5	[B]	$\text{sech}^{-1}x$
	• inverse hyperbolic cotangent			
	(code= 6)	6	[B]	$\coth^{-1}x$

Example 1:

Evaluate the following hyperbolic functions:

sinh 2.5; cosh 3.2; tanh 1.9; csch (–0.25); coth (–2.01). Also, evaluate sinh x and compare to $\frac{e^x - e^{-x}}{2}$ where $x = 1.2345 \times 10^{-8}$.

Keystrokes:	Display:	
2.5 [ENTER↑] 1 [A]	**6.0502**	sinh 2.5
3.2 [ENTER↑] 2 [A]	**12.2866**	cosh 3.2
1.9 [ENTER↑] 3 [A]	**0.9562**	tanh 1.9
.25 [CHS] [ENTER↑] 4 [A]	**-3.9586**	csch (–0.25)
2.01 [CHS] [ENTER↑]		
6 [A]	**-1.0366**	coth (–2.01)
1.2345 [EEX] [CHS] 8		
[ENTER↑] 1 [A]	**1.2345–08**	sinh (1.2345×10^{-8})
1.2345 [EEX] [CHS]		
8 [g] [e^x] [h] [LST x]		
[CHS] [g] [e^x] [–]		
2 [÷]	**1.2150–08**	incorrect due to round-off

Example 2:

Evaluate the following inverse hyperbolic functions:

$\sinh^{-1}$ (2.4); $\cosh^{-1}$ (90); $\tanh^{-1}$ (–0.65); sech^{-1} (0.4); $\coth^{-1}$ (3.4).

Keystrokes:	Display:	
2.4 [ENTER↑] 1 [B]	**1.6094**	$\sinh^{-1}$ (2.4)
90 [ENTER↑] 2 [B]	**5.1929**	$\cosh^{-1}$ (90)
.65 [CHS] [ENTER↑] 3 [B]	**-0.7753**	$\tanh^{-1}$ (–0.65)
.4 [ENTER↑] 5 [B]	**1.5668**	sech^{-1} (0.4)
3.4 [ENTER↑] 6 [B]	**0.3031**	$\coth^{-1}$ (3.4)

Polynomial Evaluation

This program performs several useful operations on polynomial functions of the form

$$f(x) = a_0 + a_1x + a_2x^2 \ldots + n_{n-1}x^{n-1} + a_nx^n$$

where the order, n, of the polynomial is 9 or less.

The following operations may be accomplished:

Evaluation of f(x):

The value, $f(x)$, may be calculated for a known value of x.

Zeros of f(x):

The real zeros of the polynomial (values of x for which $f(x) = 0$) may be found. This operation uses the SOLVE operation of the HP-34C.

Find x for a given value of f(x):

The real values of x satisfying the polynomial for a given value of $f(x)$ may be found. This operation also uses the SOLVE operation of the HP-34C.

Definite integral of f(x):

The definite integral, $\int_a^b f(x)\,dx$, of the polynomial $f(x)$, between the limits a and b may be evaluated. This operation uses the $\int_y^x$ operation built into the HP-34. (See remarks below.)

Remarks:

- This program illustrates basic but valuable ways in which the SOLVE function may be used. The program is useful for all polynomials up to order 9, having real coefficients. It can easily be expanded for use with higher order polynomials or altered for use on other types of functions.

- Since the integral of a polynomial can be readily written in closed form and easily evaluated, it is not really necessary to use the powerful INTEGRATE, $\int_y^x$, capability of the HP-34. It is used here primarily for illustrative purposes.
- The user is urged to consult the discussion in the owner's handbook on the use of the SOLVE and $\int_y^x$ capabilities for a more thorough understanding of their usefulness and limitations.
- The program will not solve for complex zeros. If it is unsuccessful in finding a real zero, **Error 4** will be displayed.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL 0	001-25, 13, 0
STO . 1	002- 23 .1
STO f I	003-23, 14, 23
h LBL 8	004-25, 13, 8
R/S	005- 74
STO f (i)	006-23, 14, 24
g DSE	007- 15 23
GTO 8	008- 22 8
R/S	009- 74
RCL . 1	010- 24 .1
STO f I	011-23, 14, 23
g R↓	012- 15 22
STO 0	013- 23 0
R/S	014- 74
h LBL A	015-25, 13, 11
f FIX 4	016- 14, 11 4
h SF 0	017-25, 51, 0
ENTER↑	018- 31
ENTER↑	019- 31
ENTER↑	020- 31
GTO 3	021- 22 3
h LBL B	022-25, 13, 12
f FIX 4	023-14, 11, 4
h SF 0	024-25, 51, 0
f SOLVE 3	025-14, 73, 3
R/S	026- 74
GTO 7	027- 22 7
h LBL 2	028-25, 13, 2

KEY ENTRY	DISPLAY
h SF 0	029-25, 51, 0
f $\int_y^x$ 3	030-14, 72, 3
R/S	031- 74
h LBL 1	032-25, 13, 1
f FIX 4	033-14, 11, 4
h CF 0	034-25, 61, 0
STO . 0	035- 23 .0
g R↓	036- 15 22
f SOLVE 3	037-14, 73, 3
R/S	038- 74
GTO 7	039- 22 7
h LBL 3	040-25, 13, 3
RCL f (i)	041-24, 14, 24
×	042- 61
g DSE	043- 15 23
h LBL 9	044-25, 13, 9
RCL f (i)	045-24, 14, 24
+	046- 51
×	047- 61
g DSE	048- 15 23
GTO 9	049- 22 9
RCL 0	050- 24 0
+	051- 51
RCL . 1	052- 24 .1
STO f I	053-23, 14, 23
g R↓	054- 15 22
h F? 0	055-25, 71, 0
h RTN	056- 25 12
RCL . 0	057- 24 .0

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
−	058- 41		

REGISTERS			I Control
R_0 a_0	R_1 a_1	R_2 a_2	R_3 a_3
R_4 a_4	R_5 a_5	R_6 a_6	R_7 a_7
R_8 a_8	R_9 a_9	$R_{.0}$ f(x)	$R_{.1}$ n
$R_{.2}$—$R_{.9}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input the order of the poly-			
	nomial (value of largest			
	exponent).	n	GSB 0	n
3	Input coefficients starting with			
	a_n	a_n	R/S	a_n
	(Repeat until all a_n through a_0			
	are entered)	a_{n-1}	R/S	a_{n-1}
		⋮		
		a_0	R/S	a_0
4	**To Evaluate the Polynomial:**			
	Input x and see $f(x)$. (Repeat			
	for all values of interest.)	x	A	$f(x)$
5	**To Find a Zero of the**			
	Polynomial:			
	Input two initial guesses in the			
	approximate range, for the	Guess 1	R/S	
	initial search.	Guess 2	B	a zero
	(Repeat for other zeros.)			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
6	**To Find x for a Given f(x):**			
	Input upper and lower guesses	Guess 1	[R/S]	
	of the expected value of x, then	Guess 2	[R/S]	
	input the known value of f(x).	f(x)	[GSB] 1	x
7	**To Find Integral of the**			
	Polynomial:			
	Set the desired display format		[f] [SCI] n	
	to control uncertainty.* Then		or, [f] [ENG] n	
	input the lower limit of the inte-	a	[ENTER↑]	
	gral and the upper limit of the	b	[GSB] 2	$\int_b^a f(x)dx$
	integral.			
	* Consult the HP-34C Owner's			
	Handbook and Programming			
	Guide for information on the			
	proper user of the display			
	format when using [∫ʸₓ]			

Example 1:

A ball is thrown straight up at a velocity of 20 meters per second, from a height of 2 meters. Neglecting air resistance, how long will it take the ball to reach the ground? The acceleration of gravity is 9.81 meters per second. From physics:

$$h(t) = x_0 + v_0 t + \frac{1}{2} at^2 = 0$$

$$= 2 + 20t + (-9.81/2)t^2 = 0$$

Keystrokes:	Display:	
2 [GSB] 0	2.0000	Input order, n
9.81 [CHS] [ENTER↑]		
2 [÷] [R/S]	-4.9050	Coefficient a_2
20 [R/S] 2 [R/S]	2.0000	Coefficients a_1, a_0
10 [ENTER↑] 0 [B]	4.1751	Seconds

There is also a negative root to this equation, but it is not relevant to this problem. To find it:

10 [CHS] [ENTER↑] 0 [B]	-0.0977	Seconds

Example 2:

The standard heat of formation of ammonia (NH_3) is given as a function of Kelvin temperature by:

$$\Delta H_T^\circ = -9140 - 7.596T + 4.243 \times 10^{-3}T^2 - 0.742 \times 10^{-6}T^3 \text{ (cal)}$$

Determine the heat of formation for temperatures of 400 K and 800 K. Also, find the temperature at which the heat of formation is -12,330.39 cal.

Keystrokes:	Display:	
3 [GSB] 0	3.0000	
.742 [CHS] [EEX]		
[CHS] 6 [R/S]	-7.4200 -07	
4.243 [EEX] [CHS]		
3 [R/S]	0.0042	
7.596 [CHS] [R/S]	-7.5960	
9140 [CHS] [R/S]	-9,140.0000	
400 [A]	-11,547.0080	ΔH°_{400}
800 [A]	-12,881.1840	ΔH°_{800}

1000 [ENTER↑]		
0 [ENTER↑]	**0.0000**	(guess for temperature)
12330.39 [CHS]		
[GSB] 1	**599.9994**	*K*

$\Delta H_T^\circ = -12{,}330.39$ calories at approximately 600 K.

Example 3:

Find the two real roots of the equation $f(x) = 4x^4 - 8x^3 - 13x^2 - 10x + 22 = 0$. Then evaluate the area under the curve (integral of $f(x)$) between the roots.

Keystrokes:	Display:	
4 [GSB] 0	**4.0000**	*n*
4 [R/S] 8 [CHS] [R/S]		
13 [CHS] [R/S]		
10 [CHS] [R/S]		
22 [R/S]	**22.0000**	coefficients input
0 [ENTER↑] 50 [B]	**0.8820**	1st real root
1 [ENTER↑] 50 [B]	**3.1180**	2nd real root
[f] [SCI] 3		
.882 [ENTER↑] 3.118		
[GSB] 2	**-7.640 01**	$\int_y^x f(x)\,dx$
[x⇄y]	**9.745 -03**	uncertainty about 1% max.

Finance

Annuities and Compound Amounts

This program can be used to solve a variety of problems involving money, time and interest. The following variables can be either inputs or outputs:

n, which is the number of compounding periods. (For a 30 year loan with monthly payments, $n = 12 \times 30 = 360$.)

i, which is the periodic interest rate expressed as a percent. (For other than annual compounding, divide the annual percentage rate by the number of compounding periods in a year, i.e., 8% annual interest compounded monthly equals 8/12 or 0.667%.)

PV, which is the present value of the cash flow or compound amounts.

PMT, which is the periodic payment.

FV, which is the future value of a compounded amount or a series of cash flows.

BAL, which is the balloon or remaining balance at the end of a series of payments.

Accumulated interest and remaining balance may also be computed with this program.

The program accommodates payments which are made at the beginning or end of compounding periods. Payments made at the end of compounding periods (ordinary annuity) are common in direct reduction loans and mortgages while payments at the beginning of compounding periods (annuity due) are common in leasing. For ordinary annuity press [A] until 1 is displayed. For annuity due press [A] until 0 is displayed.

This program uses the convention that cash outlays are input as negative, and cash incomes are input as positive.

Pressing [f] CLEAR [REG] provides a safe, convenient, easy to remember method of preparing the calulator for a new problem. However, it is not necessary to use [f] CLEAR [REG] between problems containing the same combination of variables. For instance, any number of *n*, *i*, *PV*, *PMT*, *FV* problems involving different numbers and/or different combinations of knowns could be done in succession without clearing the registers. Only the values which change from problem to problem would have to be keyed in. To change the combination of variables without using [f] CLEAR [REG], simply input zero for any variable which is no longer applicable. To go from *n, i, PMT, PV* problems to *n, i, PV, FV* problems, a zero would be stored (0 [STO] 4) in place of *PMT*. Table I summarizes these procedures.

Table I
Possible Solutions Using
Annuities and Compound Amounts

Allowable Combination of Variables	Applications: Ordinary Annuity	Applications: Annuity Due	Initial Procedure
n, i, PV, PMT (Input any three and calculate the fourth.)	Direct reduction loan Discounted notes Mortgages	Leases	Use [f] CLEAR [REG] or set *BAL* to zero
n, i, PV, PMT BAL (Input any four and calculate the fifth.)	Direct reduction loan with balloon Discounted notes with balloon	Leases with residual values	None
n, i, PMT, FV (Input any three and calculate the fourth.)	Sinking fund	Periodic savings Insurance	Use [f] CLEAR [REG] or set *PV* to zero
n, i, PV, FV (Input any three and calculate the fourth.)	Compound amount Savings (Annuity mode is not applicable and has no effect)		Use [f] CLEAR [REG] or set *PMT* to zero.

Equations:

$$-PV = \frac{PMT}{i} A\left[1-(1+i)^{-n}\right] + (BAL \text{ or } FV)(1+i)^{-n}$$

where

$$A = \begin{cases} 1 & \text{ordinary annuity} \\ (1+i) & \text{annuity due.} \end{cases}$$

Remarks:

- This program uses the [SOLVE] operation to find i. Since this is an iterative method it will take longer than the other calculations (up to 2 minutes or more). It is quite possible to define problems which cannot be solved by the technique. Such problems usually result in an error message but may simply continue to run indefinitely.
- Problems with an interest rate of 0 will give an "Error 0" display.
- Problems with *extremely* high (10^6) or low values (10^{-6}) for n or i may give invalid results.
- Interest problems with balloon payments of opposite sign to the periodic payments may have more than one mathematically correct answer (or no answer at all). This program may find one of the answers but has no way of finding or indicating other possibilities.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL A	001-25, 13, 11
f FIX 2	002-14, 11, 2
h F? 0	003-25, 71, 0
GTO 7	004- 22 7
h SF 0	005-25, 51, 0
0	005- 0
R/S	007- 74
h LBL 7	008-25, 13, 7
h CF 0	009-25, 61, 0
1	010- 1
R/S	011- 74
h LBL 1	012-25, 13, 1
0	013- 0
STO 1	014- 23 1
GSB 6	015- 13 6
RCL 5	016- 24 5
h LST x	017- 25 0
−	018- 41
RCL 3	019- 24 3
h LST x	020- 25 0
+	021- 51
÷	022- 71
CHS	023- 32
f LN	024- 14 1
RCL 6	025- 24 6
f LN	026- 14 1
÷	027- 71
STO 1	028- 23 1

KEY ENTRY	DISPLAY
h RTN	029- 25 12
h LBL 4	030-25, 13, 4
1	031- 1
STO 4	032- 23 4
GSB 6	033- 13 6
h 1/x	034- 25 2
RCL 3	035- 24 3
GSB 8	036- 13 8
×	037- 61
CHS	038- 32
STO 4	039- 23 4
h RTN	040- 25 12
h LBL 3	041-25, 13, 3
GSB 6	042- 13 6
GSB 8	043- 13 8
CHS	044- 32
STO 3	045- 23 3
h RTN	046- 25 12
h LBL 5	047-25, 13, 5
GSB 6	048- 13 6
RCL 3	049- 24 3
+	050- 51
RCL 7	051- 24 7
÷	052- 71
CHS	053- 32
STO 5	054- 23 5
h RTN	055- 25 12
h LBL 2	056-25, 13, 2
.	057- 73

KEY ENTRY	DISPLAY
2	058- 2
ENTER↑	059- 31
EEX	050- 33
CHS	061- 32
3	062- 3
h CF 1	063-25, 61, 1
f SOLVE B	064-14, 73, 12
GTO 7	065- 22 7
GTO 0	065- 22 0
h LBL 7	067-25, 13, 7
EEX	068- 33
2	069- 2
×	070- 61
STO 2	071- 23 2
R/S	072- 74
h LBL B	073-25, 13, 12
STO 8	074- 23 8
GTO 9	075- 22 9
h LBL 6	076-25, 13, 6
h SF 1	077-25, 51, 1
1	078- 1
RCL 2	079- 24 2
h %	080- 25 41
STO 8	081- 23 8
h LBL 9	082-25, 13, 9
RCL 8	083- 24 8
1	084- 1
STO 0	085- 23 0
+	086- 51

KEY ENTRY	DISPLAY
STO 6	087- 23 6
h F? 0	088-25, 71, 0
STO 0	089- 23 0
RCL 1	090- 24 1
CHS	091- 32
h y^x	092- 25 3
STO 7	093- 23 7
1	094- 1
x≷y	095- 21
−	096- 41
STO 9	097- 23 9
RCL 0	098- 24 0
×	099- 61
RCL 4	100- 24 4
RCL 8	101- 24 8
÷	102- 71
STO . 0	103- 23 .0
×	104- 61
h F? 1	105-25, 71, 1
h RTN	106- 25 12
GSB 8	107- 13 8
RCL 3	108- 24 3
+	109- 51
CHS	110- 32
h RTN	111- 25 12
h LBL 8	112-25, 13, 8
RCL 5	113- 24 5
RCL 7	114- 24 7
×	115- 61

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[+]	*116- 51*		

REGISTERS			I Unused
R_0 i or 1 + *i*	R_1 n	R_2 i(%)	R_3 PV
R_4 PMT	R_5 FV(BAL)	R_6 1 + *i*	R_7 $(1 + i)^{-n}$
R_8 i/100	R_9 $[1-(i+i)^{-n}]$	$R_{.0}$ PMT/i	$R_{.1}$ Unused
$R_{.2}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Clear the storage registers.		[f] CLEAR [REG]	
3	Set for ordinary annuity (1.00),		[A]	1.00
	or, for annuity due (0.00).		[A]	0.00
	(Press [A] until you see the			
	desired display.)			
4	Input the known values:			
	Number of periods	*n*	[STO] 1	*n*
	Periodic interest rate	*i* (%)	[STO] 2	*i* (%)
	Present value	*PV*	[STO] 3	*PV*
	Periodic payment	*PMT*	[STO] 4	*PMT*
	Future value, balloon or			
	balance	*FV (BAL)*	[STO] 5	*FV (BAL)*
5	Calculate the unknown value:			
	Number of periods		[GSB] 1	*n*
	Periodic interest rate		[GSB] 2	*i* (%)
	Present value		[GSB] 3	*PV*
	Periodic payment		[GSB] 4	*PMT*
	Future value, balloon or			
	balance		[GSB] 5	*FV (BAL)*

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
6	For a new case go to step 4 and			
	change appropriate values.			
	Input zero for any value not			
	applicable in the new case.			
7	For a new problem, go to step 2.			

Example 1:

If you place $155 in a savings account paying 5¾% compounded monthly, what sum of money can you withdraw at the end of 9 years?



Keystrokes:	Display:	
[f] CLEAR [REG]		
[A]	*0.00*	
[A]	*1.00*	ordinary annuity
9 [ENTER↑] 12 [×]		
[STO] 1	*108.00*	# of months
5.75 [ENTER↑] 12 [÷]		
[STO] 2	*0.48*	% monthly interest rate
155 [CHS] [STO] 3	*-155.00*	initial deposit
[GSB] 5	*259.74*	FV

If you desire a sum of $275 what would be the required interest rate?

275 [STO] 5	*275.00*	
[GSB] 2	*0.53*	% monthly interest rate
12 [×]	*6.39*	% yearly interest rate

Example 2:

You receive $30,000 from the bank as a 30 year, 9% mortgage. What monthly payment must you make to the bank to fully amortize the mortgage?



Keystrokes:	Display:	
[f] CLEAR [REG]		
30 [ENTER↑] 12 [×]		
[STO] 1	*360.00*	n
9 [ENTER↑] 12 [÷]		
[STO] 2	*0.75*	% monthly interest rate
30000 [STO] 3	*30,000.00*	PV
[GSB] 4	*–241.39*	PMT

Example 3:

Two individuals are constructing a loan with a balloon payment. The loan amount is $3,600 and it is agreed that the annual interest rate will be 10% with 36 monthly payments of $100. What balloon payment amount, to be paid coincident with the 36th payment, is required to fulfill the loan agreement?

(Note the cash flow diagram below is with respect to the lender. For the borrower, the appropriate diagram will be exactly the opposite.)



Keystrokes:	Display:	
[f] CLEAR [REG]		
36 [STO] 1	*36.00*	
10 [ENTER↑] 12 [÷]		
[STO] 2	*0.83*	
3600 [CHS] [STO] 3	*–3,600.00*	
100 [STO] 4 [GSB] 5	*675.27*	BAL

(Note that the final payment is $675.27 + $100.00 = $775.27 since the final payment falls at the end of the last period.)

Example 4:

This program may also be used to calculate accumulated interest/remaining balance for loans. The accumulated interest between two points in time, n_1 and n_2, is just the total payments made in that period less the principal reduction in that period. The principal reduction is the difference of the remaining balances for the two points in time. The following example demonstrates the concepts above:

For a 360 month, $50,000 loan at 9½% annual interest, find the remaining balance after the 24th payment and the accrued interest for payments 13–24 (between the 12th and 24th payments!).

First we must calculate the payment on the loan:

Keystrokes:	Display:	
[f] CLEAR [REG]		
360 [STO] 1	*360.00*	
9.5 [ENTER↑] 12 [÷]		
[STO] 2	*0.79*	
50000 [CHS] [STO] 3	*–50,000.00*	
[GSB] 4	*420.43*	PMT

The remaining balance at month 24 is:

24 [STO] 1 [GSB] 5	*49,352.76*	BAL @ month 24

Store this remaining balance and calculate the remaining balance at month 12:

[STO] [·] 1 12 [STO] 1		
[GSB] 5	*49,691.68*	BAL @ month 12

The principal reduction between payments 12 and 24 is:

[RCL] [·] 1 [–]	*338.92*	

The accrued interest is 12 payments less the principal reduction:

[RCL] 4 12 [×]	*5,045.13*	Total paid out
[x≷y] [–]	*4,706.20*	Accrued interest

Example 5:

A "third party" leasing firm is considering the purchase of a mini-computer priced at $63,000 and intends to achieve a 13% annual yield by leasing the computer to a customer for a 5-year period. Ownership is retained by the leasing firm, and at the end of the lease they expect to be able to sell the equipment for at least $10,000. What should they establish as the monthly payment in order to realize their desired yield? (Since lease payments occur at the start of the periods, this is an annuity due problem.)



Keystrokes:	Display:	
[f] CLEAR [REG]		
[A]	*0.00*	annuity due
5 [ENTER↑] 12 [×]		
[STO] 1	*60.00*	
13 [ENTER↑] 12 [÷]		
[STO] 2	*1.08*	
63000 [CHS] [STO] 3	*–63,000.00*	
10000 [STO] 5	*10,000.00*	
[GSB] 4	*1,300.16*	

If the price is increased to $70,000 what should the payments be?

70000 [CHS] [STO] 3	*–70,000.00*	
[GSB] 4	*1,457.73*	

If the payments were increased to $1,500 what would the yield be?

1500 [STO] 4	*1,500.00*	
[GSB] 2	*1.18*	% per month
12 [×]	*14.12*	% per year

Discounted Cash Flow Analysis

Two forms of discounted cash flow analysis are the net present value *(NPV)* approach and the internal rate of return *(IRR)* approach. This program calculates either *NPV* or *IRR* for up to 8 groups of cash flows.

The amount of the initial investment is input (observing the sign convention) followed by the positive or negative amounts of each group of future cash flows, and the number of times the series of cash flows occurs. The cash flows must occur at equal intervals.

After the initial investment and all cash flows have been entered, the user may: 1) Input an assumed interest rate and calculate the net present value *(NPV)* of the investment, or 2) Calculate the internal rate of return *(IRR)*. The *IRR* is an interest rate that equates the present value of a set of cash flows with an initial investment. It is the interest rate that is obtained when the calculated net present value of a series of cash flows is zero. *IRR* is also called the *yield* or *discounted rate of return*.

This program uses the convention that cash outlays are input as negative, and cash incomes are input as positive.

Remarks:

- Calculation of *IRR* may take several minutes (5 or more) depending on the number of cash flow entries.
- The cash flow sequence (including the initial investment) must contain at least *one* sign change.
- Cash flows with multiple sign changes may have multiple answers. This program may find one answer but has no way of indicating other possibilities.
- The program is designed for optimum operation when 0% < *IRR* ≤ 100%. It often will solve for interest rates outside these ranges but will display *Error* messages for zero interest rates or if it is unable to find a solution.

KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000-
[h] [LBL] [A]	001-25, 13, 11
[STO] [.] 8	002- 23 .8
1	003- 1
[.]	004- 73
0	005- 0
1	006- 1
6	007- 6
[STO] [f] [I]	008-23, 14, 23
[x⇆y]	009- 21
[h] [LBL] 0	010-25, 13, 0
[R/S]	011- 74
[x⇆y]	012- 21
[STO] [f] [(i)]	013-23, 14, 24
[g] [R↓]	014- 15 22
[g] [ISG]	015- 15 24
[STO] [f] [(i)]	016-23, 14, 24
[g] [ISG]	017- 15 24
[GTO] 0	018- 22 0
[R/S]	019- 74
[h] [LBL] 1	020-25, 13, 1
[h] [SF] 1	021-25, 51, 1
[EEX]	022- 33
2	023- 2
[÷]	024- 71
[h] [LBL] [B]	025-25, 13, 12
[RCL] [f] [I]	026-24, 14, 23
[h] [INT]	027- 25 32
1	028- 1

KEY ENTRY	DISPLAY
[−]	029- 41
[STO] [f] [I]	030-23, 14, 23
[STO] [.] 7	031- 23 .7
[g] [R↓]	032- 15 22
[h] [F?] 1	033-25, 71, 1
[GTO] 2	034- 22 2
1	035- 1
[ENTER↑]	036- 31
[EEX]	037- 33
[CHS]	038- 32
3	039- 3
[f] [SOLVE] 2	040-14, 73, 2
[GTO] 3	041- 22 3
[GTO] 9	042- 22 9
[h] [LBL] 3	043-25, 13, 3
[EEX]	044- 33
2	045- 2
[×]	046- 61
[R/S]	047- 74
[h] [LBL] 2	048-25, 13, 2
1	049- 1
[+]	050- 51
[STO] 0	051- 23 0
0	052- 0
[h] [LBL] 4	053-25, 13, 4
[RCL] 0	054- 24 0
[RCL] [f] [(i)]	055-24, 14, 24
[CHS]	056- 32
[h] [y^x]	057- 25 3

KEY ENTRY	DISPLAY
[×]	058- 61
1	059- 1
[h] [LST x]	060- 25 0
[−]	061- 41
[g] [DSE]	062- 15 23
[RCL] [f] [(i)]	063-24, 14, 24
[×]	064- 61
[+]	065- 51
[g] [DSE]	066- 15 23
[GTO] 4	067- 22 4

KEY ENTRY	DISPLAY
[RCL] 0	068- 24 0
1	069- 1
[−]	070- 41
[÷]	071- 71
[RCL] [·] 8	072- 24 .8
[+]	073- 51
[RCL] [·] 7	074- 24 .7
[STO] [f] [I]	075-23, 14, 23
[g] [R↓]	076- 15 22
[h] [CF] 1	077-25, 61, 1

REGISTERS			I Control
R_0 $1+i$	R_1 CF_1	R_2 N_1	R_3 CF_2
R_4 N_2	R_5 CF_3	R_6 N_3	R_7 CF_4
R_8 N_4	R_9 CF_5	$R_{.0}$ N_5	$R_{.1}$ CF_6
$R_{.2}$ N_6	$R_{.3}$ CF_7	$R_{.4}$ N_7	$R_{.5}$ CF_8
$R_{.6}$ N_8	$R_{.7}$ Used	$R_{.8}$ CF_0	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Optional: set display status			
	as desired. (suggest FIX 2.)			
3	Input the initial investment.*	INV	[A]	INV
4	Beginning with the first period,			
	key in each cash flow* and the			
	number of times it occurs,			
	pressing [R/S] after each	CF	[ENTER↑]	
	group. (Be sure to key in 1 if	N	[R/S]	N
	the cash flow occurs only			
	once.)			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
5	Repeat step 4 until all cash			
	flows have been input. Then go			
	to step 6 for IRR or step 7 for			
	NPV.			
6	TO CALCULATE INTERNAL			
	RATE OF RETURN:		[B]	*IRR* (%)
7	TO CALCULATE NET			
	PRESENT VALUE:			
	Input the applicable periodic			
	interest (discount) rate.	i (%)	[GSB] 1	*NPV*
8	Always return to step 3 for a			
	new case or a new calculation.			
	* Be sure to observe proper sign			
	convention for initial investment			
	and cash flows.			

Example 1:

An investor pays $80,000 for a duplex that he intends to sell after 7 years. He must spend some money the first year for repairs. At the end of the seventh year the duplex is sold for $91,000. Will he achieve a desired 9% after tax yield with the following after tax cash flows?



Keystrokes:	Display:
[f] [FIX] 2	
80000 [CHS] [A]	-80,000.00
600 [CHS] [ENTER↑]	
1 [R/S]	1.00
6500 [ENTER↑] 1 [R/S]	1.00
8000 [ENTER↑] 2 [R/S]	2.00
7500 [ENTER↑] 2 [R/S]	2.00
91000 [ENTER↑] 1 [R/S]	1.00
9 [GSB] 1	-4,108.06

Since the NPV is negative the investment does not achieve the desired 9% yield.

Example 2:

An investment of $620,000,000.00 is expected to have the following annual income stream for the next 15 years:

Number of Years:	Cash Flow ($):
First 10 years	100,000,000
next 5 years	5,000,000

What is the expected rate of return?

Keystrokes:	Display:	
620000000 [CHS] [A]	-620,000,000.0	
100000000 [ENTER↑]		
10 [R/S]	10.00	
5000000 [ENTER↑]		
5 [R/S]	5.00	
[B]	10.06	(annual IRR of 10.06%)

Recreation

Moon Rocket Lander

Imagine for a moment the difficulties involved in landing a rocket on the moon with a strictly limited fuel supply. You're coming down tail-first, freefalling toward a hard rock surface. You'll have to ignite your rockets to slow your descent; but if you burn too much too soon, you'll run out of fuel 100 feet up, and then you'll have nothing to look forward to but cold eternal moon rocks coming faster every second. The object, clearly, is to space your burns just right so that you will alight on the moon's surface with no downward velocity.

The game starts off with the rocket descending at a velocity of 50 feet/second from a height of 500 feet. The velocity and altitude are shown in a combined display as -50.0500, the altitude appearing to the right of the decimal point and the velocity to the left, with a negative sign on the velocity to indicate downward motion. Then the remaining fuel is displayed and the rocket fire count-down begins: "3", "2", "1", "0",. Exactly at zero you may key in a fuel burn. You only have one second, so be ready. A zero burn, which is very common, is accomlished by doing nothing. After a burn the sequence is repeated unless:

1. You have successfully landed—flashing zeros.
2. You have smashed into the lunar surface—flashing crash velocity.

You must take care, however, not to burn more fuel than you have; for if you do, you will free-fall to your doom! The final velocity shown will be your impact velocity (generally rather high). You have 60 units of fuel initially.

Equations:

We don't want to get too specific, because that would spoil the fun of the game; but rest assured that the program is solidly based on some old friends from Newtonian physics:

$$x = x_0 + V_0 t + \frac{1}{2} at^2, \quad V = V_0 + at, \quad V^2 = V_0^2 + 2a(x - x_0)$$

where:

x, V, a, and t are distance, velocity, acceleration, and time.

Remarks:

- Only integer values for fuel burn are allowed. [R/S] can be used to stop Moon Rocket Lander at any time.

KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	000-
[h] [LBL] [A]	001- 25, 13, 11
5	002- 5
0	003- 0
0	004- 0
[STO] 6	005- 23 6
5	006- 5
0	007- 0
[CHS]	008- 32
[STO] 7	009- 23 7
6	010- 6
0	011- 0
[STO] 8	012- 23 8
[h] [LBL] 0	013- 25, 13, 0
[RCL] 6	014- 24 6
[f] [FIX] 4	015- 14, 11, 4
[EEX]	016- 33
4	017- 4
[÷]	018- 71
[RCL] 7	019- 24 7
[h] [ABS]	020- 25 34
[+]	021- 51
[RCL] 7	022- 24 7
[g] [x>0]	023- 15 51
[GSB] 4	024- 13 4
[x⇄y]	025- 21
[CHS]	026- 32
[h] [PSE]	027- 25 74
[h] [PSE]	028- 25 74

KEY ENTRY	DISPLAY
[f] [FIX] 0	029- 14, 11, 0
[RCL] 8	030- 24 8
[h] [PSE]	031- 25 74
3	032- 3
[h] [PSE]	033- 25 74
2	034- 2
[h] [PSE]	035- 25 74
1	036- 1
[h] [PSE]	037- 25 74
0	038- 0
[h] [PSE]	039- 25 74
[h] [LBL] 9	040- 25, 13, 9
[RCL] 8	041- 24 8
[x⇄y]	042- 21
[f] [x>y]	043- 14 51
[GTO] 6	044- 22 6
[STO] [−] 8	045- 23, 41, 8
2	046- 2
[×]	047- 61
5	048- 5
[−]	049- 41
[STO] 9	050- 23 9
2	051- 2
[÷]	052- 71
[RCL] 6	053- 24 6
[+]	054- 51
[RCL] 7	055- 24 7
[+]	056- 51
[RCL] 9	057- 24 9

KEY ENTRY	DISPLAY
STO + 7	058- 23, 51, 7
g R↓	059- 15 22
STO 6	060- 23 6
h INT	061- 25 32
g x>0	062- 15 51
GTO 0	063- 22 0
RCL 7	064- 24 7
h LBL 7	065- 25, 13, 7
h PSE	066- 25 74
GTO 7	067- 22 7
h LBL 6	068- 25, 13, 6
RCL 8	069- 24 8
2	070- 2
•	071- 73
5	072- 5
−	073- 41
STO + 6	074- 23, 51, 6

KEY ENTRY	DISPLAY
2	075- 2
×	076- 61
STO + 7	077- 23, 51, 7
RCL 6	078- 24 6
1	079- 1
0	080- 0
×	081- 61
RCL 7	082- 24 7
g x^2	083- 15 3
+	084- 51
f $\sqrt{x}$	085- 14 3
CHS	086- 32
GTO 7	087- 22 7
h LBL 4	088- 25, 13, 4
x⇄y	089- 21
CHS	090- 32
x⇄y	091- 21

REGISTERS			I Unused
R_0	R_1	R_2	R_3
R_4	R_5	R_6 X	R_7 V
R_8 FUEL	R_9 ACCEL.	$R_{.0}$—$R_{.6}$ Unused	

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Assume manual control.		A	"V.ALT"
				"FUEL"
				"3"
				"2"

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
				"1"
				"0"
3	Key in burn:			
	Upon "0" display,			
	press R/S then		R/S	
	enter burn.	BURN	R/S	"V.ALT"
				"FUEL"
				"3"
				"2"
				"1"
				"0"
4	Go to step 3 until you land			
	(flashing zeros) or crash			
	(flashing impact velocity).			
5	If you survived last landing			
	attempt, go to step 2 for another			
	try.			

Nimb

The game of Nimb begins with a collection of N objects, or as the calculator plays it, with the positive number N. Each player alternately subtracts one, two, or more from the total until only one is left. The player forced to take the last one loses.

To begin the game you specify the maximum number that can be taken in a single move. Then you tell the calculator how many objects you wish to start with (i.e., the value of N).

After each move the machine will display the remaining total. A negative sign indicates that it is the player's move next, while a positive display indicates that it is the HP-34C's move.

As the challenger you are allowed to make the first move. It is possible for you to win, but of course the HP-34C is a master player: it will not let you make an error and win. If you cheat by taking more than the specified limit the calculator will catch you and force you to repeat the move.

This program is based on an HP-25 program by James L. Horn.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL A	001-25, 13, 11
f FIX 0	002 14, 11, 0
STO 0	003- 23 0
1	004- 1
+	005- 51
STO 1	006- 23 1
3	007- 3
5	008- 5
0	009- 0
7	010- 7
.	011- 73
1	012- 1
STO 2	013- 23 2
5	014- 5
5	015- 5
1	016- 1
7	017- 7
8	018- 8
STO 3	019- 23 3
RCL 0	020- 24 0
h LBL 1	021-25, 13, 1
R/S	022- 74
h LBL B	023-25, 13, 12
f FIX 0	024-14, 11, 0
STO 0	025- 23 0
h LBL 4	026-25, 13, 4
CHS	027- 32
R/S	028- 74

KEY ENTRY	DISPLAY
+	029- 51
g x<0	030- 15 41
GTO 0	031- 22 0
RCL 3	032- 24 3
GTO 1	033- 22 1
h LBL 0	034-25, 13, 0
h LST x	035- 25 0
1	036- 1
f x>y	037- 14 51
GTO 2	038- 22 2
g R↓	039- 15 22
RCL 1	040- 24 1
f x≤y	041- 14 41
GTO 2	042- 22 2
x⇄y	043- 21
STO − 0	044-23, 41, 0
RCL 0	045- 24 0
R/S	046- 74
1	047- 1
−	048- 41
RCL 1	049- 24 1
÷	050- 71
h FRAC	051- 25 33
RCL 1	052- 24 1
×	053- 61
g x=0	054- 15 71
1	055- 1
STO − 0	056-23, 41, 0
h LBL 2	057-25, 13, 2

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
RCL 0	058- 24 0	RCL 2	061- 24 2
g x≠0	059- 15 61	f FIX 1	062- 14, 11, 1
GTO 4	060- 22 4	GTO 1	063- 22 1

REGISTERS			I Unused
R_0 Total	R_1 Max+1	R_2 3507.1	R_3 55178
R_4—$R_{.9}$ Unused			

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Indicate the maximum number			
	of objects which can be re-			
	moved in one move.	MAX	A	MAX
3	Indicate the total number of			
	objects with which you wish to			
	start the game, (usually 15).	N	B	-N
4	If the number in the display is			
	negative, key in your move and			
	see the number remaining.	MOVE	R/S	+REM
5	If the number in the display is			
	positive, let the HP-34C move.		R/S	-REM
6	Do steps 4 and 5 until the			
	game is over.			
7	At the end of the game turn the			
	calculator upside down to read			
	the message.			
	If calculator loses:			I LOSE
	If calculator wins:			BLISS

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
8	For another game:			
	If max. move remains same, go			
	to step 3.			
	For different max. move go to			
	step 2.			

Example:

Starting with 15 objects, with a maximum allowable move of 3, play Nimb with the HP-34C.

Keystrokes:	Display:	
3 A	3.	
15 B	-15.	Ready
3 R/S	12.	Player takes 3
R/S	-9.	HP-34C takes 3
5 R/S	-9.	Player tries to cheat.
2 R/S	7.	Player takes 2
R/S	-5.	HP-34C takes 2
3 R/S	2.	Player takes 3
R/S	-1.	HP-34C takes 1
1 R/S	55178.	Player takes last one and loses.

Turn calculator upside down for message: BLISS

General

Timers

This program converts your HP-34C into a timer which can operate as 1) a count-down timer, counting down to display zero when the preset time has elapsed, or 2) a count-up timer, displaying the elapsed time since the timer was started. The upper limit of the count-up timer is approximately 10 minutes.

When using this program, you should remember that the clock circuits of the HP-34C are designed for calculator use, not for accurate time-keeping. Although the routine may be calibrated quite accurately, highly stable performance should not be expected.

Equation:

$$C_{\text{NEW}} = C_{\text{OLD}} \frac{\text{HP time}}{\text{Actual time}}$$

Remarks:

- All times are input or displayed in (H.MMSS) format.
- Your calibration constants may differ substantially from those given here.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL A	001-25, 13, 11
h CF 0	002-25, 61, 0
f FIX 4	003-14, 11, 4
STO 0	004- 23 0
h LBL 9	005-25, 13, 9
0	006- 0
R/S	007- 74
STO 1	008- 23 1
g →H	009- 15 6
RCL 0	010- 24 0
×	011- 61
h INT	012- 25 32
STO f I	013-23, 14, 23
RCL 1	014- 24 1
R/S	015- 74
h LBL 8	016-25, 13, 8
g DSE	017- 15 23
GTO 8	018- 22 8
GTO 9	019- 22 9
h LBL B	020-25, 13, 12
h SF 0	021-25, 51, 0
f FIX 4	022-14, 11, 4
STO 2	023- 23 2
h LBL 6	024-25, 13, 6
•	025- 73
9	026- 9
9	027- 9
9	028- 9

KEY ENTRY	DISPLAY
STO f I	029-23, 14, 23
0	030- 0
R/S	031- 74
h LBL 7	032-25, 13, 7
g ISG	033- 15 24
GTO 7	034- 22 7
h LBL B	035-25, 13, 12
RCL f I	036-24, 14, 23
h INT	037- 25 32
RCL 2	038- 24 2
÷	039- 71
f →H.MS	040- 14 6
R/S	041- 74
GTO 6	042- 22 6
h LBL 1	043-25, 13, 1
g →H	044- 15 6
x⇄y	045- 21
g →H	046- 15 6
x⇄y	047- 21
−	048- 41
RCL 1	049- 24 1
g →H	050- 15 6
÷	051- 71
h 1/x	052- 25 2
h F? 0	053-25, 71, 0
GTO 0	054- 22 0
RCL 0	055- 24 0
×	056- 61
R/S	057- 74

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
GTO A	058- 22 11	×	061- 61
h LBL 0	059- 25, 13, 0	R/S	062- 74
RCL 2	060- 24 2	GTO B	063- 22 12

REGISTERS			I Counter
R_0 C_d	R_1 Time	R_2 C_u	R_3—$R_{.9}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
	COUNT-DOWN TIMER:			
2	Input count-down timer			
	constant (try 5600).	C_D	A	0.0000
3	Input desired time.	T (H.MMSS)	R/S	T (H.MMSS)
4	Start timer.		R/S	
5	Timer display blinks until			
	0.0000 is displayed and time			
	has elapsed.			0.0000
6	For a new time, T, go to step			
	3. To recalibrate, go to			
	step 12.			
	COUNT-UP TIMER:			
7	Input count-up timer constant			
	(try 5400).	C_U	B	0.0000
8	Start timer.		R/S	
9	After desired period has			
	elapsed, stop timer.		R/S	0.0000
10	Display the elapsed time.		B	T (H.MMSS)

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
11	Reset the timer and go to step			
	8 for another run. To recalibrate		R/S	0.0000
	the timer, go to step 14.			
12	To calibrate the count-down			
	timer, input the ending time	T_e(H.MMSS)	ENTER↑	
	and the starting time and	T_s(H.MMSS)	GSB 1	C_D NEW
	calculate new constant.			
13	To proceed, press R/S			
	and go to step 3.		R/S	0.0000
14	To calibrate the count-up timer,			
	store the displayed elapsed	T (H.MMSS)	STO 1	T (H.MMSS)
	time. Then input the ending	T_e(H.MMSS)	ENTER↑	
	time and the starting time.	T_s(H.MMSS)	GSB 1	C_U NEW
15	To proceed, press R/S and			
	go to step 8.		R/S	0.0000

Example 1:

Use the count-down time to measure an elapsed timer of 35 seconds, and another of 1 minute, 8 seconds. Try an initial calibration constant of 5600.

Keystrokes:	Display:	
5600 A	*0.0000*	
.0035 R/S	*0.0035*	Count-down time
R/S	*0.0000*	Time has elapsed

Suppose you recorded, from your watch, an ending time of 3:42:56 and a starting time of 3:42:23, i.e., 33 seconds elapsed. Recalibrate the timer with this information.

3.4256 ENTER↑		
3.4223 GSB 1	*5939.3937*	C_D NEW
R/S	*0.0000*	

Now count-down for 1 minute, 8 seconds.

.0108 [R/S]	***0.0108***
[R/S]	***0.0000***

The true elapsed time should now be quite close to that input. If it is not, recalibrate again.

Operate the count-up timer for a known elapsed time (say 40 seconds) and recalibrate as needed. Use an initial calibration constant of 5400.

Keystrokes:	**Display:**	
5400 [B]	***0.0000***	
[R/S]		start timer
[R/S]	***0.0000***	stop timer
[B]	***0.0039***	elapsed time

True elapsed time was 40 seconds. Recalibrate:

[STO] 1	***0.0039***	
.004 [ENTER↑]	***0.0049***	
0 [GSB] 1	***5,265.0000***	$C_{U\ NEW}$ (you may get a different number).
[R/S]	***0.0000***	Input new constant and reset counter

Random Number Generator

Random numbers are useful in a wide variety of applications, for example: in simulation, sampling, computer programming, numerical analysis and games.

This program includes routines to calculate:

1. Uniformly distributed pseudo-random numbers in the range $0 \leqslant r \leqslant 1$.
2. Uniformly distributed integers from 0 to 9.

It also includes a routine to simulate dealing from an infinite deck of cards. The Ace is 1; the Jack, 11; the Queen, 12; the King, 13. All other cards count their face value.

The random number generator:

$$r_{n+1} = \text{FRAC}(9821 \times r_n + .211327)$$

passes the spectral test (Knuth, V.2, no. 3.4) and, because its parameters satisfy Theorem A (op.cit., p. 15) it generates one million distinct random numbers between 0 and 1 regardless of the value selected for r_0. This generator was developed by Don Malm as part of an HP-65 Users' Library program.

KEY ENTRY	DISPLAY
f CLEAR PRGM	000-
h LBL A	001-25, 13, 11
f FIX 4	002-14, 11, 4
STO 0	003- 23 0
GSB 0	004- 13 0
h LBL 3	005-25, 13, 3
GSB 9	006- 13 9
R/S	007- 74
GTO 3	008- 22 3
h LBL B	009-25, 13, 12
f FIX 0	010-14, 11, 0
STO 0	011- 23 0
GSB 0	012- 13 0
h LBL 4	013-25, 13, 4
GSB 9	014- 13 9
1	015- 1
0	016- 0
×	017- 61
h INT	018- 25 32
R/S	019- 74
GTO 4	020- 22 4
h LBL 1	021-25, 13, 1
f FIX 0	022-14, 11, 0
STO 0	023- 23 0
GSB 0	024- 13 0
h LBL 5	025-25, 13, 5
GSB 9	026- 13 9
1	027- 1
3	028- 3

KEY ENTRY	DISPLAY
×	029- 61
1	030- 1
+	031- 51
h INT	032- 25 32
R/S	033- 74
GTO 5	034- 22 5
h LBL 0	035-25, 13, 0
9	036- 9
8	037- 8
2	038- 2
1	039- 1
STO 1	040- 23 1
.	041- 73
2	042- 2
1	043- 1
1	044- 1
3	045- 3
2	046- 2
7	047- 7
STO 2	048- 23 2
h RTN	049- 25 12
h LBL 9	050-25, 13, 9
RCL 0	051- 24 0
RCL 1	052- 24 1
×	053- 61
RCL 2	054- 24 2
+	055- 51
h FRAC	056- 25 33
STO 0	057- 23 0

		REGISTERS	I Unused
R_0 r_i	R_1 9821	R_2 .211327	R_3—$R_{.9}$ Unused

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input a seed: any number			
	between 0 and 1.	SEED		
	For random numbers go to step			
	3. For random integers go to			
	step 5. For a simulated card			
	dealer go to step 7.			
3	To generate a random number			
	$(0 \leq r \leq 1)$		A	r_1
4	For another random number.		R/S	r_2
	Repeat step 4 as desired for			
	further random numbers.			
5	To generate random integers			
	$(0 \leq INT \leq 9)$.		B	INT_1
6	For another random integer.		R/S	INT_2
	(Repeat step 6 as desired for			
	further random integers.)			
7	To deal cards:		GSB 1	$CARD_1$
8	For further cards.		R/S	$CARD_2$
	(Repeat step 8 as desired to			
	deal more cards.)			
9	For a new start, go to step 2 to			
	input a new seed as desired.			

Example 1:

Using an initial seed of 0.2356 generate a string of pseudo-random numbers.

Keystrokes:	**Display:**
.2356 [A]	*0.0389*
[R/S]	*0.5134*
[R/S]	*0.2538*
etc.	

Example 2:

Using an initial seed of .12345 deal a hand of 5 cards.

Keystrokes:	**Display:**	
.12345 [GSB] 1	*8.*	
[R/S]	*2.*	
[R/S]	*1.*	Ace
[R/S]	*9.*	
[R/S]	*11.*	Jack

Moving Average

In a moving average, a specified number of data points is averaged. When there is a new piece of input data, the oldest piece of data is discarded to make room for the latest input. This replacement scheme makes the moving average a valuable tool in following trends. The fewer the number of data points, the more trend-sensitive the average becomes. With a large number of data points, the average behaves more like a regular average, responding slowly to new input data.

This program allows for a moving average span of 1 to 17 units. The number of units, n, must be specified before any data input begins by keying it in and pressing [B] . Then the data is input by keying in each value, x_k, and pressing [A] in turn. The calculator will display the current input number, k, until at least n values have been entered. After the n^{th} value (and for all succeeding values), the calculator will flash the current input number before halting with the moving average, AVG, in the display.

The value of the average may be displayed at any time by pressing [GSB] 1. This feature allows the average to be calculated before n data points have been input. The average is based on the number of inputs or n, whichever is smaller.

Remarks:

This program clears all registers during its execution. Therefore any data stored in extra registers will be destroyed.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	*000-*	[g] [DSE]	*023- 15 23*
[h] [LBL] [B]	*001-25, 13, 12*	[GTO] 5	*024- 22 5*
[f] [FIX] 2	*002-14, 11, 2*	[RCL] [·] 8	*025- 24 .8*
[f] CLEAR [REG]	*003- 14 33*	[STO] [f] [I]	*026-23, 14, 23*
[STO] [·] 8	*004- 23 .8*	[h] [LBL] 5	*027-25, 13, 5*
[STO] [f] [I]	*005-23, 14, 23*	[g] [R↓]	*028- 15 22*
[R/S]	*006- 74*	[h] [RTN]	*029- 25 12*
[h] [LBL] [A]	*007-25, 13, 11*	[h] [LBL] 0	*030-25, 13, 0*
[RCL] [·] 9	*008- 24 .9*	[x⇄y]	*031- 21*
1	*009- 1*	[h] [PSE]	*032- 25 74*
[+]	*010- 51*	[RCL] 0	*033- 24 0*
[x⇄y]	*011- 21*	[RCL] [·] 8	*034- 24 .8*
[RCL] [f] [(i)]	*012-24, 14, 24*	[÷]	*035- 71*
[STO] [−] 0	*013-23, 41, 0*	[ENTER↑]	*036- 31*
[x⇄y]	*014- 21*	[h] [RTN]	*037- 25 12*
[STO] [f] [(i)]	*015-23, 14, 24*	[h] [LBL] 1	*038-25, 13, 1*
[STO] [+] 0	*016-23, 51, 0*	[RCL] 0	*039- 24 0*
[g] [R↓]	*017- 15 22*	[RCL] [·] 9	*040- 24 .9*
[x⇄y]	*018- 21*	[RCL] [·] 8	*041- 24 .8*
[STO] [·] 9	*019- 23 .9*	[f] [x≤y]	*042- 14 41*
[RCL] [·] 8	*020- 24 .8*	[x⇄y]	*043- 21*
[f] [x≤y]	*021- 14 41*	[g] [R↓]	*044- 15 22*
[GSB] 0	*022- 13 0*	[÷]	*045- 71*

REGISTERS			I Control
R_0 Σ	R_1 Used	R_2 Used	R_3 Used
R_4 Used	R_5 Used	R_6 Used	R_7 Used
R_8 Used	R_9 Used	$R_{.0}$ Used	$R_{.1}$ Used
$R_{.2}$ Used	$R_{.3}$ Used	$R_{.4}$ Used	$R_{.5}$ Used
$R_{.6}$ Used	$R_{.7}$ Used	$R_{.8}$ n	$R_{.9}$ K

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input number of points in the			
	average, $(1 \leq n \leq 17)$.	n	[B]	n
3	Input data point and compute			
	moving average. *	x_k	[A]	"k", AVG
4	Optional: Display average at			
	any time.		[GSB] 1	AVG
5	For a new case, go to step 2.			
	* If an error is made in data			
	input, you must start over.			
	The average is not displayed			
	until after the n^{th} point is			
	input.			

Example 1:

A six-period moving average is used to project monthly sales. The first 6 months of sales are as follows:

Month	1	2	3	4	5	6
Sales	125	183	207	222	198	240

Compute the moving average. Also compute the average after month three.

Keystrokes:	**Display:**	
6 [B]	*6.00*	
125 [A]	*1.00*	
183 [A]	*2.00*	
207 [A]	*3.00*	
[GSB] 1	*171.67*	(average after month three)
222 [A]	*4.00*	
198 [A]	*5.00*	
240 [A]	*"6.00", 195.83*	

The actual sales for the seventh month totalled 225 units. Calculate a new moving average.

225 [A]	*"7.00", 212.50*	